

```

1  ! subroutine Compute_Solution_* ( T, A, Z, N0, N_T, B )
2
3      real(rk), intent(in) :: T          ! Independent variable
4      real(rk), intent(in) :: A(:, :)    ! Coefficient matrix for ODE
5      real(rk), intent(in) :: Z(:, :)    ! Coefficient matrix solutions
6      real(rk), intent(in) :: N0(:)      ! Initial condition
7      real(rk), intent(out) :: N_T(:)    ! Solution N(T)
8      real(rk), intent(in), optional :: B(:) ! For inhomogeneous equations.
9      ! Let M be the index in A of the last nonzero diagonal element.
10     ! Then B(1:m) is B in d N(t)/d t = A ( N(t) + B )
11     ! and B(m+1:) is C in d N(t)/d t = A N(t) + C. Remember that here,
12     ! columns of A after M are zero, so this is just a quadrature.
13     ! Backsolve_A_C_B can compute B(1:m) if all you have is C.
14     real(rk) :: E
15     integer :: I, J, M, R, U
16
17     u = ubound(a,1)
18
19     ! Find the first zero diagonal in A.
20     ! Assume A(i,j) = 0 for i>j if A(j,j) = 0.
21     do m = 0, u - 1
22         if ( a(m+1,m+1) == 0 ) exit
23     end do
24
25     if ( .not. present(b) ) then

```

Compute the homogeneous solution for the first  $m$  equations.

$$N_i(t) = N_i(0) e^{a_{ii}t} + \sum_{j=1}^{i-1} z_{ij} (N_j(t) - N_j(0) e^{a_{jj}t}) \quad i \leq m \quad (1)$$

```

31     do i = 1, m
32         e = exp(a(i,i) * t)
33         N_T(i) = n0(i) * e + &
34             & sum ( [ ( z(i,j) * ( N_T(j) - n0(j) * e ), j = 1, i-1 ) ] )
35     end do
36
37     if ( m < u ) then

```

Compute the integrals of the first  $m$  equations:

$$\int_0^t N_l(t) dt = N_l(0) \frac{e^{a_{ll}t} - 1}{a_{ll}} + \sum_{j=1}^{l-1} z_{lj} \left( \int_0^t N_j(t) dt - N_j(0) \frac{e^{a_{jj}t} - 1}{a_{jj}} \right), \quad l \leq m. \quad (2)$$

```

45
46     block

```

```

47      real(rk) :: N_Int(m)
48      do i = 1, m
49          e = ( exp(a(i,i) * t) - 1 ) / a(i,i)
50          n_int(i) = n0(i) * e + &
51              & sum( [ ( z(i,j) * ( n_int(j) - n0(j) * e ), j = 1, i-1 ) ] )
52      end do

```

Compute the homogeneous solution for the remaining  $m + 1 : u$  equations:

$$N_i(t) = \sum_{l=1}^m a_{il} \int_0^t N_l(t) dt + N_i(0), i > m \quad (3)$$

```

59
60      n_t(m+1:u) = matmul(a(m+1:u,1:m), n_int(1:m))+ n0(m+1:u)
61  end block
62  end if
63  else

```

Compute the inhomogeneous solution for the leading  $m$  equations,

$$N_i(t) = \hat{N}_i(0) e^{a_{ii}t} + \sum_{j=1}^{i-1} z_{ij} \left( \hat{N}_j(t) - \hat{N}_j(0) e^{a_{jj}t} \right) - B_i, i \leq m, \quad (4)$$

where  $\hat{\mathbf{N}}(0) = \mathbf{N}(0) + \mathbf{B}$ .

```

71      do i = 1, m
72          e = exp(a(i,i)*t)
73          N_T(i) = ( n0(i) + b(i) ) * e - b(i) + &
74              & sum ( [ ( z(i,j) * ( N_T(j) + b(j) - ( n0(j) + b(j) ) * e ), &
75                  & j = 1, i-1 ) ] )
76      end do
77
78      if ( m < u ) then

```

Compute the integrals of the first  $m$  equations:

$$\int_0^t N_l(t) dt = \hat{N}_l(0) \frac{e^{a_{ll}t} - 1}{a_{ll}} + \sum_{j=1}^{l-1} z_{lj} \left( \int_0^t \hat{N}_j(t) dt - \hat{N}_j(0) \frac{e^{a_{jj}t} - 1}{a_{jj}} \right) - B_l t, l \leq m, \quad (5)$$

where  $\hat{\mathbf{N}}(0) = \mathbf{N}(0) + \mathbf{B}$ .

```

88      block
89          real(rk) :: N_Int(size(n_t))
90          do i = 1, m
91              e = ( exp ( a(i,i) * t ) - 1 ) / a(i,i)
92              n_int(i) = ( n0(i) + b(i) ) * e - b(i) * t + &
93                  & sum ( [ ( z(i,j) * ( n_int(j) + b(j)*t - &
94                      & ( n0(j) + b(j) ) * e ), j = 1, i-1 ) ] )
95          end do

```

Compute the inhomogeneous solution for the remaining  $n - m$  equations:

$$N_i(t) = \sum_{l=1}^m a_{il} \int_0^t N_l(t) dt + C_i t + N_i(0), i > m. \quad (6)$$

Here,  $C_i$  is in the array B.

```

102         n_t(m+1:u) = matmul(a(m+1:u,1:m), n_int(1:m)) + b(m+1:u)*t + n0(m+1:u)
103     end block
104 end if
105 end if
106
107 ! end subroutine Compute_Solution_*
```